Mathematics 244: Lab 4 Systems of First Order Linear Differential Equations

In this lab we use Maple to find eigenvalues and eigenvectors of matrices, and solve linear systems of algebraic equations and systems of first order linear differential equations. We also obtain pictures of the slope fields of these equations in the phase plane.

Please turn in only the printout of your Maple worksheet. Include explicit answers to all questions asked, using the **text** feature of Maple to insert them in the worksheet. Remove from the worksheet any extraneous material and any errors you have made.

Before you start up Maple, first copy the seed file into your directory from the Web page of the course.

Please turn in only the printout of your Maple worksheet. Include explicit answers to all questions asked, using the **text** feature of Maple to insert them in the worksheet. Remove from the worksheet any extraneous material and any errors you have made.

In addition to the `DEtools` and `plots` libraries used in other labs, we require a Linear Algebra package. There are now two such packages in Maple, not completely compatible. The newer package, called `LinearAlagebra`, introduced in Maple6 is used here since it has more respect for its user. In **Section 0**, the libraries are loaded and two matrices are defined.

```
with(LinearAlgebra):with(DEtools):with(plots):
A:= «-1,8>|<4,-5>|<1,10»;
B:= «1|2>,<2|-2>,<2|1»;
```

Section 1. Matrix Operations. We first try some operations. Some will give errors; some may give unexpected results. **The errors in this section should be left in the worksheet** since they illustrate the the definition of the operations. Consult the help pages to find properties of the operations used in these statements and **write a brief description of all operators used here** (you may need to use a **Full Text Search** for "dot" to find all the help pages for that operator).

```
A+B;
A+1;
A.A;
(2).A;
M1:=A.B;
M2:=B.A;
(Vals1,Vecs1):=Eigenvectors(M1);
M1ˆ(-1);
(Vals2,Vecs2):=Eigenvectors(M2);
C:=«1|1|0>,<0|1|0>,<0|0|2»;
Eigenvectors(C);
```

The relation found here between the eigenvalues of `M1` and `M2` is quite general, although the usual proof would be a distraction in this course. In special cases, though, it follows from the easy observation that left multiplication by B takes an eigenvector of AB to an eigenvector of BA and left multiplication by A takes an eigenvector of BA to an eigenvector of AB.

The discussion section of this part should describe the operations on vectors used here, including their limitations. It should also note the significance of a zero column in the matrix of eigenvectors of C.

Section 2. Real matrix Exponentials. The `LinearAlgebra` package provides a good interface to numerical work with matrices, but it needs to be cajoled into doing symbolic work. This can be done with the `map` function. (There is also a function called `Map` that modifies `Matrices` in place, but we need to construct a new matrix without affecting the original one, so we will use the lower case variant.)

When eigenvalues of an n by n matrix M are real and distinct, the eigenvectors form a **basis** of the space of $\mathbb{R}^n$. For each eigenvalue λ_i, the corresponding eigenvector v_i is the vector of coefficients of $e^{\lambda_i t}$ in a solution of $d\mathbf{y}/dt = M\mathbf{y}$. Since the exponential factor takes the value 1 when $t = 0$, the initial conditions give a system of equations whose matrix of coefficients Φ whose i^{th} column is the eigenvector v_i of M and whose right side is $\mathbf{y}(0)$. The solution $\Phi^{-1}\mathbf{y}(0)$ is the vector of coefficients of the special solutions $\mathbf{v}_i e^{\lambda_i t}$. This leads to the expression

$$\mathbf{y} = \Phi e^{\Lambda t} \Phi^{-1} \mathbf{y}(0),$$

where $e^{\Lambda t}$ is a diagonal matrix whose entries are the functions $e^{\lambda_i t}$.

The construction of this matrix when M is the 3×3 matrix `M2` from Section 1 uses the following Maple commands (requiring results found in Section 1) that are included in the seed file.

```
EL2:=DiagonalMatrix(map(c->exp(c*t),Vals2));
Y2:=Vecs2.EL2.Vecs2^(-1);
DY2:=map(diff,Y2,t);
MY2:=M2.Y2;
Equal(DY2,MY2);
subs(t=0,Y2);
```

In the **discussion** of this section, indicate how to use this work to solve

$$\frac{d\mathbf{y}}{dt} = M_2\mathbf{y} \quad \text{with} \quad \mathbf{y}(0) = \begin{bmatrix} 3 \\ -2 \\ 1 \end{bmatrix}$$

(where M_2 is the matrix denoted `M2` in the worksheet). Then find some initial conditions so that the equation has a constant solution and verify that these constants satisfy the equation.

Section 3. Saddle points and nodes. Consider the matrices

$$M_{3A} = \begin{bmatrix} 3 & -2 \\ 5 & -4 \end{bmatrix} \quad \text{and} M_{3B} = \begin{bmatrix} 4 & 1 \\ -2 & 1 \end{bmatrix}.$$

For each, we will use the method of Section 2 (which will be a little easier here because these systems have 2×2 matrices) to solve the equation $d\mathbf{y}/dt = M\mathbf{y}$ with initial conditions

$$(a) \quad \mathbf{y}(0) = \begin{bmatrix} 1 \\ 0 \end{bmatrix}, \qquad (b) \quad \mathbf{y}(0) = \begin{bmatrix} -1 \\ 1 \end{bmatrix}, \qquad (c) \quad \mathbf{y}(0) = \begin{bmatrix} -1 \\ -1 \end{bmatrix}.$$

A graphical check of the solution involves plotting the slope field of the equation in a **phase plane** whose coordinates are the components of $\mathbf{y}$ and superimposing a **parametric plot** of the **trajectories** of the solutions. Because the equations are **autonomous**, all solutions starting at a point on one of these trajectories will follow the trajectory — the only difference being the value of t at which it visits a particular point. This uses features not explored previously in this course, so the seed file contains instructions to construct the individual plots for the first of these equations. You will need to combine these plots with a `display` command and repeat the process for the second equation. Here are the instructions that appear in the seed file.

```
M3A:=«3|-2>,<5|-4»;
(Vals3A,Vecs3A):=Eigenvectors(M3A);
EL3A:=DiagonalMatrix(map(c->exp(c*t),Vals3A));
Y3A:=Vecs3A.EL3A.Vecs3A^(-1);
Y3Aa:=Y3A.<1,0>;
Y3Ab:=Y3A.<-1,1>;
Y3Ac:=Y3A.<-1,-1>;
VecVar:=<y1(t),y2(t)>;
listVar:=convert(VecVar,'list');
eq3A:=[diff(VecVar[1](t),t)=(M3A.VecVar)[1],
       diff(VecVar[2](t),t)=(M3A.VecVar)[2]];
range3:=y1=-2..2,y2=-2..2;
Field3A:=DEplot(eq3A,listVar,t=-1..1,range3,color=BLACK):
Sol3B:=plot([[Y3Aa[1],Y3Aa[2],t=-1..1],
             [Y3Ab[1],Y3Ab[2],t=-1..1],
             [Y3Ac[1],Y3Ac[2],t=-1..1]],range3):
display({Field3A,Sol3A},title="Equation 3A");
```

We need $y_1(t)$ and $y_2(t)$ to be combined into a `Vector` in order to use the `LinearAlgebra` package, and into a `list` to serve as a argument of the `DEplot` function. The `convert` instruction assures that the related objects will have the same contents in different formats. The entries of the vector appear in the differential equation because the `diff` operation only applies to scalar functions. The expression `Sol3A` plots a **list** of objects, each of which is a parametric description of the trajectory of a solution in the phase plane.

You should construct another chain of statements to produce a similar plot for the second equation.

The **Discussion** portion of this section should investigate the role of these plots in checking this method of solving differential equations. In particular, do the claimed solution curves look like they follow the slope field of the equation? For which of these equations has is the origin a saddle point? Does the shape of the solution agree with what you expect. For the other equation, is the origin **stable** (i.e., attracting) or **unstable** (i.e., repelling)? Describe how both the eigenvalues of the coefficient matrix and the slope field illustrate your classification. For both equations, find the solutions whose trajectories lie along straight lines.

Section 4. Spiral points. Consider the equation

$$\frac{dY}{dt} = \begin{bmatrix} 3 & 5 \\ -2 & 1 \end{bmatrix} Y. \tag{S}$$

Supplementary notes for this course show that, if M is a 2×2 real matrix with eigenvalues $a \pm bi$ and J is defined by $M = aI + bJ$, then

$$e^{Mt} = e^{at}\big((\cos bt)I + (\sin bt)J \big).$$

The seed file implements this solution in Maple for (S), leading to a matrix `Y4`. You should use methods explored in previous sections to **verify** that this matrix is e^{Mt}, and to **illustrate** the solutions with the same initial values (a), (b) and (c) used in Section 3 superimposed on a slope field of this equation.

Here are the instructions leading to Y4.

```
M4:=«3|5>,<-2|1»;
E4:=Eigenvalues(M4);
(a4,b4):=Re(E4[1]),Im(E4[1]);
J4:=(1/b4).(M4-a4);
Y4:=Multiply(exp(a4*t),Multiply(cos(b4*t),IdentityMatrix(2))
                      +Multiply(sin(b4*t),J4));
```

Section 5. Repeated Eigenvalues. A similar process to the one used in Section 4 can be applied in the case of a matrix like

$$M_5 = \begin{bmatrix} -1 & -2 \\ 18 & 11 \end{bmatrix}$$

that has a repeated eigenvalue. If M is a 2×2 matrix with a as a double eigenvalue, then $M = aI + N$ where N^2 is the zero matrix. General properties of matrix exponentials show that

$$e^{Mt} = e^{aIt}e^{Nt} = e^{at}(I + Nt). \qquad (N)$$

You don't need to **derive** this result to show that it gives the solution. Instead, you can modify the instructions of Section 4 to implement the expression for $e^{M_5 t}$ given by (N), obtaining a matrix Y5. **Verify** that the derivative of Y5 is equal to the matrix obtained by multiplying on the left by M_5, and that Y5 reduces to the identity matrix when $t = 0$.

Then, **illustrate** these solutions with a plot of the direction field of the equation in the phase plane with the solutions with the initial conditions (a), (b) and (c) introduced in Section 3.

End of Lab 4